


6 Automatic Testing

6.1 Using Macros to Automate Testing


Production testing requires only that a PASS/FAIL result be obtained. You may set up CableEye so that the operator needs only to press the TEST pushbutton on the fixture and read results on nearby LED lamps. Using the mouse or keyboard, or reading the screen, becomes unnecessary. A *Macro* lets you do this by performing a sequence of steps *automatically*. Each step represents a basic operation that you could perform manually, if desired, such as "Test Cable", or "Print Difference List". Other operations valid only in a Macro let you issue warning tones, create loops, or post messages for the operator to see.

The test engineer would initially create a Macro to achieve the desired operation. To begin testing, the operator loads the Macro and presses the TEST button – no further knowledge of the computer or set-up procedure is necessary.


1 – Click the Macro button to open the "Automatic Test" window. Then choose the Macro to load from the menu. You may create any number of Macros for different jobs. You will find the example shown here on your distribution disk.




2 – The Macro specifies the test operations you need for a particular job. Easily create Macros by just choosing commands from a menu (described later). You may modify the Macro at any time with the "Edit" function, and save the final Macro on disk for future use.




3 – Press the TEST pushbutton on the fixture, or click the "Execute" button on the screen, to begin testing. Results are shown on bright LED lamps. The PASS lamp is *green*, and the FAIL lamp is *red*.




4 – If CableEye detects a fault, this Macro shows the problem in a "Differences List" (Line 10 of the Macro). The problem found here is a missing connection between pin 15 on the left and pin 15 on the right.



5 – If the cable matches exactly, a large green box with a white checkmark appears, and the error list contains no entries. To see a graphical representation of wiring faults, click the "View Wiring" button just above the +/- column.



6 – Macros can perform many functions and are not limited in length. We will review this and other Macros later in this section.



Information about MACROS

Quick Key (none)

Execute Time Depends on action selected

Test Fixture Required

Special Screens (none)

LED Lamps Depends on selected functions

Usable in Macro No (cannot call a Macro from within a Macro)

Related Disk File <MACRONAME> .MAC

Possible Messages ...

"Executing Macro. Press ESC to stop" shown during the execution of a Macro when no errors are encountered."

"Macro error. You used an IF statement without first using COMPARE TEST TO MATCH."

"No Macro instructions entered. Please choose one or more instructions before saving."

"Macro error. No match file found. Please check filename and correct your Macro" when LOAD cannot find the specified file.

"Macro has been changed. Discard changes?"

"Macro Error: Indented instructions must be preceded by an IF statement."


Effect: The Macro function in CableEye allows you to automate a test sequence for repetitive cable tests or frequently used test procedures. In addition, the system may be set up so that an unskilled operator may simply turn on the computer and move directly to the test fixture to check a cable, using only the TEST pushbutton to trigger a test operation. The use of Macros ensures test consistency from cable to cable, preserves the test procedure for future use, and eliminates operator error.

6.2 The Macro Window

A special window provides all the controls you need to create, select, and execute Macros. Click the "Macro Window" button to turn this window on, and click it again, or click the Close box, to turn it off.




When the Macro window appears, the data windows you had been viewing (such as "Test Data Wiring", "Match Data Wiring", or "Wiring Differences", right) temporarily close. In their place appears your choice of data windows for Macro Execution (whatever you last selected while in the Macro mode, lower right). These data windows could be the same as those you had been using before you opened the Macro window, or they could be different. In any case, when you close the Macro window, your original selection of data windows before opening Macros reappears. In this way, you may configure the view that suits your needs for batch testing without affecting the way you use CableEye while testing individual cables. Any changes in the data windows made by the Macro itself, such as executing the instruction `DISPLAY DIFFERENCE LIST`, appear while in the Macro mode only and will not affect the window arrangement you had been using before opening the Macro window.



Open
Macro
Window




Close
Macro
Window



Remember that holding the SHIFT key while you click "View Wiring" or "View Netlist" leaves the original window open and creates a new one, rather than replacing one with the other. This lets you view three or more windows simultaneously, a feature that may be helpful when using a large-screen monitor.


6.3 Executing Macros

The Macro window appears as seen on the right when open. The text box under "Macro Name" shows the currently selected Macro; click the *Down-Arrow Tab* to see the entire selection of available Macros and choose the one you need. With the Macro name highlighted as shown here, you may also use the *Up-* and *Down-Arrow Keys* on the keyboard to scroll through the list of Macros.



The *Description* field is an editable text block. Add comments freely. Any text you enter here is saved automatically with the Macro for future reference.

Execute – Click the "Execute" button, press the TEST pushbutton on the fixture, or depress the optional footpedal (Item 714), to start the Macro you see listed in the *Test Sequence* text block. The instructions execute automatically, one after the other, until a WAIT, PAUSE, STOP, or END instruction is reached. The WAIT FOR PB ("Wait for Pushbutton") instruction creates a pause until you press the black TEST pushbutton on the fixture, click the "Continue" button on the screen, or press the ENTER key on the keyboard, allowing time to disconnect one cable and connect the next. Understand that depressing the footpedal is electrically equivalent to pressing the TEST button and will not be given explicitly as an option every time.




The "Execute" button changes to "Pause" after you begin. Click "Pause" to temporarily stop the Macro without losing the next instruction address or clearing the loop counter. Once paused, the button changes to "Resume" which you may click when ready to continue testing. During a pause, you may test an individual cable, load data from the database, or print a report. Clicking "Resume" continues the execution where you left



off rather than starting over from the beginning. Note that if you turn off the Macro mode during a Pause, the next instruction address and loop counter will be cleared. Be sure to restore the Macro's needed Match data, if changed, before resuming execution.

Continue – The "Continue" button remains dimmed unless the Macro is expecting a User cue (*Example: WAIT FOR PB instruction*). If the Macro is waiting, click "Continue", press the TEST pushbutton, or press the ENTER key to continue testing after the next cable has been attached to the fixture. Click "Stop" to end execution and reset the next instruction address to "1".




Step – Click the "Step" button instead of "Execute" to process instructions one at a time. When one instruction finishes, the software always pauses and will wait until you press "Step" again before continuing. In this way, you may study the effect each instruction has before going further, an especially useful tool when debugging a new Macro. Once you are convinced that your Macro performs as intended, click "Execute" to continue at full speed.

Close Log – Use the "Close Log" button to manually close a log file that had been opened within a Macro. Normally, you would have a Close Log instruction within the Macro. However, if you pause or stop a Macro before Close Log is reached and do not wish to resume, you must use "Close Log" to manually close the disk file that had been accepting test results. If you attempt to exit CableEye without closing the log file, you will see a warning message, and if you proceed to exit, CableEye will close the log automatically. However, once a log file is closed, it cannot be reopened, and any new test results must be placed in a new file.

6.4 Creating and Editing Macros

At least twelve example Macros come with the CableEye software to illustrate some of the ways you can employ automatic testing. These examples are *ready to use* and cover many typical situations, so you may not need to create anything new. However, there may be circumstances where you might like to do something differently than we have done with these examples. This section explains how to create and edit a Macro. If possible, it would be helpful if you had CableEye operating while you read this so you could work through the examples as we discuss them.

Two buttons located at the bottom of the Macro window activate the Macro editing function. They remain dimmed during Macro execution and become active only when execution stops.




Edit – Click the "Edit" button to add or remove instructions in an existing Macro. When you are finished editing and save your changes, the altered Macro will replace the original one.

New – Click the "New" button to open the edit window with a completely blank slate. From this screen, you may create a new Macro of your choice. When finished editing, click "Save" to write the new file to disk.

The Macro Edit Window – Clicking either "Edit" or "New" changes the layout of the window to that shown on the right. In this case, we clicked "New" to start entering a new Macro.

You see a list of permissible Macro commands on the left and your actual Macro sequence on the right. You will select instructions from the command list one at a time to assemble the Macro sequence you need.




Scroll down the command list to review the instructions. Note that most of the operations you've previously used when testing individual cables, like "Test Cable", appear in the list. You will also see special commands useful only in a Macro, like "If Match Then . . ."

Macro Files – Macros are stored as individual files in the "Macros" folder located within your CableEye folder. When creating a new Macro, enter a name of your choice in the "Macro Name" text box. You may use names greater than the DOS eight-character limitation, if desired. All Macro files end in ".MAC" to identify them to the CableEye software, and must be located in the Macros folder to be recognized. There is no real limit to the number of Macros you can have. To delete a Macro, you must drag the Macro file into the trash from the Windows desktop. To ensure that removal is a deliberate operation, we do not provide a "Delete" button in the Macro window.


Example – We will develop a simple Macro to illustrate the procedure. As we go along, we will offer explanations for your review. If possible, work along with the book on your computer; this will reinforce the ideas we discuss.

1 – If you have not already done so, click the "New" button and type in the name "EXAMPLE" for the Macro's name (as shown on the previous page).

2 – Click in the text entry box under "Command List". Then, type the letter "L" to see all commands that begin with that letter. Double-click on LEARN CABLE to add it to the test sequence. Note that you may either scroll to a command or continue typing its name until no other selections are shown.




3 – Continue selecting Macro commands and adding them to the list until the Macro has the five instructions plus END as shown at the right.




The next instruction we need is BEEP, shown selected on the right but not yet entered in the list. Because we want to hear a tone *only* if the cable passes and not otherwise, we need to make the BEEP instruction dependent

on the IF MATCH THEN . . . being *true*. We do this by *indenting* the BEEP instruction in the list. To indent an instruction, *type a space in the command list before selecting it!* Look at the example above and note that a space appears before the "B". Double click on BEEP now and it will become an indented instruction, executing only when the IF MATCH THEN . . . statement is true.


4 – Enter additional instructions so the Macro appears as shown on the right. Here, we have made an intentional mistake by not indenting the BEEP BEEP instruction. To correct this, we must first delete the unindented BEEP BEEP. Do so by clicking on the errant instruction, at which time it will become **bold**, and then pressing the DELETE key on the keyboard. Now you may reenter the instruction with the proper indent.



5 – The last instruction in this Macro creates a loop to line 3. This will bring us back to the WAIT instruction and allows the test sequence to repeat executing after you load the next cable.




Select REPEAT in the list and press ENTER (or double click on it). A small window titled "Macro Parameter" appears, at which time you should type in the line number to which you will jump.



You may jump *forward* instead of backward by entering a line number greater than that of the REPEAT instruction. If the destination instruction has not yet been entered, you may need to enter a temporary value and go back to change it later. What do you think will happen if you enter the line number of the repeat instruction itself? Try it and see! (click "Stop" to end execution).

6 – Save your work by clicking the "Save" button. If you pressed "Cancel", all changes you made would be discarded and nothing saved to disk. Click the "Print" button, just to the left of "Save", to print hard copy of this Macro on your report printer.

7 – To test the Macro, first mount a cable of your choice in the test fixture. Then click "Execute" to begin. When execution starts, the message bar at the bottom of the screen will turn red and show the current Test Count and Error Count. You will also note that the instruction currently executing will appear **bold**.




In this Macro, CableEye will learn the attached cable as "good", sound a tone, and then wait for you to press the TEST pushbutton on the fixture. When ready, press the button to run through a typical test cycle. To force an error, loosen the cable slightly at one end to create a few open circuits, and press TEST again. Note that a different tone sounds for the error condition. Click "Stop" to stop execution.

Important: The Test Count value normally starts at "1" and increments *automatically* when the REPEAT instruction executes. You may embed the Test Count in Notes or Label text to create serial numbers, preset the Count value to an initial value at the beginning of a Macro using the "SET INITIAL COUNT =" instruction, and create a conditional statement based on the count value using the "IF COUNT =" instruction. We discuss this in more detail later.


8 – Stop the Macro, and try starting again by clicking "Step" instead of "Execute". In this case, each instruction waits for you to click "Step" again before executing. You will find this very helpful when checking long Macros or when it is not responding as you expect.




9 – To add a new instruction to the Macro, click EDIT to return to the familiar Macro edit window. Let's display a wiring graphic showing open circuits if an error occurs. To do so, click on the instruction in the Test Sequence *just below* where you want the display operation to occur – that instruction will then show as **bold**. In this case, it should be "REPEAT 3" (right). Then, from the Command List text box, type a space to create an indent, and select DISPLAY OPENS.



10 – Save the edited Macro. The final result should appear as shown at the right.



11 – Run the modified Macro and check it by learning the cable you used before, and then loosening it in the fixture to cause an open circuit. This time, when you test it, you should see the Open Circuits graphic window (right); your errors may be different than the ones shown here. Click the "View Shorts" button if you suspect there may be shorts also.



6.5 Macro Editing Summary

This list summarizes some important facts about Macro editing:

Insertion Point: The bold instruction in the list indicates the insertion point. New instructions will be inserted *above* this. Change the insertion point by clicking on a different instruction while in the Edit mode.

```
8 IF NOMATCH THEN...
9 BEEP BEEP
10 *END*
```

Deleting Instructions: Click on the instruction you wish to delete and press the DELETE key on the keyboard, or CTRL-X ("Cut to Clipboard") while in the Edit mode.

Inserting Instructions: Place the insertion point and either select the desired instruction from the list, or press CTRL-V ("Paste from Clipboard") while in the Edit mode.

Conditional Instructions: Indented instructions are *conditional* and will be executed only if the previous "IF"-type instruction is true. Create an indented instruction by first typing a space in the Command List's text box, and then choosing the instruction while in the Edit mode. You may have as many indented instructions as you wish following an IF statement.


```
4 TEST CABLE
5 COMPARE TEST TO MATCH
6 IF MATCH THEN...
7 BEEP
8 IF NOMATCH THEN...
9 BEEP BEEP
10 DISPLAY OPENS
11 REPEAT 3
12 *END*
```

Branch Instructions: The "REPEAT n" instruction transfers control to a new line number. Usually we branch back to repeat a test cycle, but you may also branch forward by entering a line number greater than that of the Repeat instruction.


```
7 BEEP
8 IF NOMATCH THEN...
9 BEEP BEEP
10 DISPLAY OPENS
11 REPEAT 3
12 *END*
```

To enter a line number, select REPEAT and press ENTER (or double click on it) while in the Edit mode. A pop-up window will appear into which you will enter the destination line number.

Printing Macros: Click the print button visible in the Edit window. You must open the Edit window to access this button.



Macro Names: All Macros have an extension of ".MAC" and are stored in the "Macros" folder within your CableEye folder.



Deleting Macros: Drag the Macro you wish to delete from the Macros folder into the trash.

6.6 Macro Instructions Explained

Forty-one Macro instructions appear in the command list at the time of this writing. We typically add new instructions as the software advances, so you may find more than this number in the software you receive (go to the Customer Services area of our web site to download updated documentation, www.camiresearch.com). The command list you see in Edit mode presents these instructions in alphabetical order, although all are easily accessed by typing the first few letters of the instruction, and then scrolling into the shortened list to pick what you need.

Some of the instructions function exactly like their corresponding commands in the main function list (for example, TEST CABLE). Others apply only in Macros to control the flow of a Macro sequence, provide cues for the operator, and monitor the TEST pushbutton on the test fixture (for example, WAIT FOR PB). In this section, we explain the Macro instructions for your reference, and group them functionally rather than alphabetically. Note that we discuss *all data logging commands* (there are eight of them) in Section 7, *Data Logging*, not here.

Cable Measurement Commands

TEST CABLE – measures a cable and copies data into the Test Data buffer. Operates exactly like the "Test Cable" button.

LEARN CABLE – measures a cable and copies data into the Match Data buffer. Operates exactly like the "Learn Cable" button.

CONTINUOUS TEST – measures a cable continuously while looking for changes in continuity. When this executes during a Macro, you should press the TEST pushbutton when ready to stop the measurement and advance to the next instruction. In a Macro, the intermittent connections error display appears during testing and disappears when you go on to the next instruction. CONTINUOUS TEST asserts an error condition if any changes in continuity occur, so you may follow with IF MATCH THEN... or IF NOMATCH THEN... for conditional operations.

Note that CONTINUOUS TEST *clears the Test Data buffer* when it executes. Thus, if you plan on doing additional tests on the attached cable after you finish checking for intermittent connections but before looping back, be sure to execute TEST CABLE again to reread the cable.

Database Commands

LOAD . . . – copies a specific cable file from the database into the Match Data buffer during Macro execution. When you add this instruction to the sequence during editing, a selection list appears showing all available file names in the database. From this list, you must choose one. Selecting the file copies your choice into the Macro Sequence, with the exact file name you chose embedded in the instruction (for example, LOAD DB9M-DB9F-S9D). When executed, this instruction always loads the specified file. This is very similar to LOAD MATCH DATA in the main function list, but with the cable choice predetermined.

ENTER CABLE NAME – displays a window *when executed* asking the operator to enter a cable name. When the operator chooses a file, the selected cable file is copied into the Match Data buffer. Unlike **LOAD...**, no specific file is connected with this instruction. Rather, it is a "run-time" load in that the operator is asked to select the desired file at the time of execution.

SEARCH DISK FOR MATCH – searches the database for an identical match based on the contents of the Test Data buffer. Connector types, genders, and wiring must all match. If the software finds an identical match, it copies that cable file into the Match Data buffer, turns on the PASS LED, and deasserts the error condition. Anything other than an identical match will leave the Match Data buffer unchanged, turn on the FAIL LED, and assert an error. Because an error condition is asserted upon mismatch, you may follow SEARCH DISK FOR MATCH with IF MATCH THEN... or IF NOMATCH THEN... for conditional operations.

Note 1: If multiple matches are found, a pop-up window will appear asking you to choose which one to load. Although the wiring will be the same for both, the Descriptive Notes and Label Text may be different, and this would affect printed output should the Macro later print reports or labels.

Note 2: Caution! If a defective cable, because of its defect, matches another cable in the database, SEARCH DISK FOR MATCH will cause that incorrect cable to load and no error will be asserted.

Note 3: If no Test Data is present when SEARCH DISK FOR MATCH executes, an error message will appear and the Macro will be stopped; this would represent a Macro design error.

Control Commands

WAIT FOR PB (wait for pushbutton) – pauses execution indefinitely until you press the TEST pushbutton on the fixture, depress the optional footswitch, click the "Continue" button on the screen, or press the ENTER key on the keyboard. Use this instruction to allow time for the operator to remove a cable that has been tested and mount a new one before allowing execution to continue.

WAIT FOR PB . . . <userfile.nte> (conditional wait for pushbutton) – pauses execution indefinitely while displaying a message of your choice in a pop-up window along with three control buttons (see test sequence at right and pop-up window below right). To resume normal execution, click the first button on the left (here labeled "Continue"), press the TEST pushbutton on the fixture, depress the optional footswitch, or press the ENTER key on the keyboard.



To execute the indented commands following this instruction, click the middle button (here labeled "Change Cable"). Clicking the right button always pauses the Macro. Use this instruction instead of the standard WAIT FOR PB if you wish to change the Match data or otherwise alter the operation of the Macro based on the operator's decision *without restarting the Macro*. For example, you may change cable types within a Macro without disrupting data logging.

To set up this instruction, you must create a note file, naming it as you wish and entering text of your choice. On the first line of the file text, enter the label you want to appear for the *left* button, and on the second line, enter text for the *middle* button. Your message begins on the third line and may be as long as necessary. The note file on the right will produce the pop-up window you see above it.

```

Test Sequence
1  ENTER CABLE NAME
2  BEEP
3  WAIT FOR PB...change.nte
4  ENTER CABLE NAME
5  TEST CABLE
6  COMPARE TEST TO MATCH
7  IF MATCH THEN...
8  BEEP
9  IF NOMATCH THEN...
10 BEEP BEEP
11 REPEAT 3
12 *END*

```



Your note file must have an extension ".NTE" and be located in the "Notes" folder within your CableEye folder. When you choose this instruction during creation of your Macro, a selection list appears showing all available note file names (files ending in .NTE) in your "Notes" folder. From this list, you must choose one. Clicking OK copies your choice into the Macro Sequence, with the exact file name you chose embedded in the instruction (for example, NOTE FILE . . . change.nte).

COMPARE TEST TO MATCH – compares Test Data to Match Data. If Test and Match Data are exactly equal in wiring, connector type, and gender, the error condition is *deasserted* (meaning that there are *no faults*). Anything other than exact equality asserts the fault condition. This instruction sets the error state prior to using IF MATCH THEN... or IF NOMATCH THEN...

IF MATCH THEN . . . – executes *indented* commands that follow, provided the match condition is *true* (error condition *deasserted*). Should the match condition be false (error condition *asserted*), execution will skip to the next *unindented* command. Only SEARCH DISK FOR MATCH, COMPARE TEST TO MATCH, and CONTINUOUS TEST affect the error flag, so one of these instructions must precede IF MATCH THEN . . .; no other commands have any effect on the error flag.

IMPORTANT: When you want to subject commands to a preceding IF statement, you must *indent* them. Do so by *typing a space in the text box before selecting the command* from the command list. Indented commands that do not follow an IF statement or another indented command will produce an error message and stop the Macro.

IF NOMATCH THEN . . . – executes *indented* commands that follow, provided the match condition is *false* (error condition *asserted*). This command functions just like IF MATCH THEN . . ., except that the match condition is *inequality*.

IF COUNT = n – executes *indented* commands that follow, provided the loop counter equals "n". You would enter the value "n" at the time you create the Macro, and this value would then be shown embedded in the instruction (for example, IF COUNT = 45). "n" may be any numeric value from 1 through 9999.

SET INITIAL COUNT = n – fixes the starting value of the loop-counter at the value "n". You would enter the value "n" at the time you create the Macro and this value would then be shown embedded in the instruction (for example, **SET INITIAL COUNT = 108**). "n" may be any numeric value from 1 through 9999.

ENTER INITIAL COUNT – displays a window asking the operator to enter a starting loop-counter value. When the operator enters a numeric value and clicks OK or presses ENTER, the entered value becomes the current loop-counter value. Valid entries range between 1 and 9999. Unlike **SET INITIAL COUNT =**, no specific count value is connected with this instruction. Rather, it is a "run-time" entry in that the operator is asked to select the desired count at the time of execution. Use this when creating serial numbers based on the loop-counter and starting at a value other than "1".

SKIP CABLE COUNT – prevents the loop-counter from advancing when REPEAT executes. Use this when creating serial numbers, or during data logging, to freeze the loop-counter if you encounter a defective cable. This may also be helpful when jumping *forward* using the REPEAT instruction.

PAUSE n – inserts a pause of "n" seconds in the command flow without having any other effect. You may wish to use PAUSE in place of WAIT FOR PB to insert a fixed delay while a test cable is being changed. You would enter the value "n" at the time you create the Macro, and this value would then be shown embedded in the instruction (for example, **PAUSE 5**). "n" may be any numeric value from 1 through 9999.

REPEAT n – transfers execution to line "n". In most cases, line "n" is smaller than the current line, causing a branch back to a previous instruction. However, you may jump forward if desired by inserting a line number that is greater than the current line. You would enter the value "n" at the time you create the Macro and this value would then be shown embedded in the instruction (for example, **REPEAT 3**). "n" may be any line number on which you've entered, or will enter, an instruction. REPEAT n automatically increments the loop-counter, regardless of whether you jump backward or forward. Use SKIP CABLE COUNT prior to REPEAT to leave the loop-counter unaffected.

BEEP – sounds Window's *asterisk* tone through the sound card on your computer. If you do not have a sound card installed, you should hear a single beep tone on the computer's built-in speaker. Use this to announce successful completion of a test, or to cue the Operator to mount another cable.

BEEP BEEP – sounds Window's *critical stop* tone through the sound card on your computer. If you do not have a sound card installed, you should hear two successive beep tones on the computer's built-in speaker. Use this to warn that an error or mismatch has occurred.

STOP – *terminates* a Macro and returns control to the Macro menu. You may also press ESC during Macro execution to achieve the same effect.

NOTE FILE . . . <userfile.nte> – pauses execution and displays a specific text file containing instructions or comments you have written for the operator. When you choose this instruction during creation of your Macro, a selection list appears showing all available note file names (files ending in .NTE) in your "Notes" folder (located within your main CableEye folder). From this list, you must choose one. Clicking OK copies your choice into the Macro Sequence, with the exact file name you chose embedded in the instruction (for example, **NOTE FILE . . . next . nte**). When executed, an "Operator Memo" window appears with your text shown. The notes window remains visible until you press the TEST pushbutton, click OK in the pop-up window, or press ENTER on the keyboard. At that time the Macro resumes. You may use NOTE FILE... in place of WAIT FOR PB if you wish to show a message at the beginning of each test loop.

Create your .NTE file using Word Pad or any word processor of your choice. Be sure the file has an extension of ".NTE", and place it in the "Notes" folder within your main CableEye folder. The displayed notes will automatically word-wrap to fill the text box, so you need not insert carriage returns (line breaks) when you are entering the note text. Enter two successive carriage returns to start a new paragraph. If you choose a note file during Macro creation and later delete the actual file from your directory without updating the Macro, the NOTE FILE instruction will be skipped during execution. See also "WAIT FOR PB . . ." for a conditional WAIT that shows a text box like "NOTE FILE . . ." but has two buttons you may label yourself, each offering a different branch outcome.

Display Commands

DISPLAY MATCH NOTES – activates the Match Notes tab in the lower left corner of the screen, rendering the Match Data notes visible. If they were already visible, this instruction has no effect.

DISPLAY WIRING – turns on the Test Data Wiring window and immediately continues to the next instruction. The new selection of display windows is automatically refitted to the available display size. Use "WAIT FOR PB", "NOTE FILE . . .", or "PAUSE n" to pause execution, if desired, after this instruction executes. The Window will remain visible until it is replaced by another type of display window. The software allows only the Automatic Test window and one display window during Macro execution when using DISPLAY instructions in a Macro. However, you may customize the window display in advance of starting the Macro if you want three or more windows to appear simultaneously. In this case, you may not want to use DISPLAY WIRING within the Macro as it will reformat the display area for the Automatic Test window and the Wiring Display only, clearing your custom arrangement.

DISPLAY NETLIST – turns on the Test Data Netlist window and immediately continues to the next instruction. Comments given above for DISPLAY WIRING apply to this instruction also.

DISPLAY DIFFERENCE LIST – turns on the Differences List window and immediately continues to the next instruction. Comments given above for DISPLAY WIRING apply to this instruction also.

DISPLAY OPENS – turns on the Differences List window with the Missing Connections graphic active, and immediately continues to the next instruction. Comments given above for DISPLAY WIRING apply to this instruction also.

DISPLAY SHORTS – turns on the Differences List window with the Extra Connections graphic active, and immediately continues to the next instruction. Comments given above for DISPLAY WIRING apply to this instruction also.

Printing Commands

PRINT MATCH DATA SPECS – prints the currently loaded Match Data specifications on your report printer, including the title block with your custom company information, wiring schematic, wire list, descriptive notes, and label text. This functions exactly like clicking the print button. If one page is insufficient, additional pages will be generated as necessary. Printing requires about 15 seconds per page on a PCL-5 or above printer. During printing, you should see an activity light flash on your printer. The computer's date and time at the time of printing appear in the title block. Date and time printing may be suppressed if desired (see page 11-8).



PRINT TEST DATA SPECS – (operates just like PRINT MATCH DATA SPECS, except that it prints the Test Data specifications). To use this instruction in a Macro, you must manually enter Test Data label text before starting the Macro. If you execute Test Cable within the Macro, previously entered Test Notes or Label text will be cleared.

PRINT DIFFERENCE LIST – prints the difference list on your report printer. This functions exactly like printing the difference list manually from the difference list screen. Because COMPARE TEST TO MATCH generates the difference list, it must be executed first. If Test and Match data are identical, nothing is printed and execution continues without stopping. If no valid comparison data is present, an error message appears and a tone sounds, but the Macro continues executing. If one page is insufficient, additional pages will be generated as necessary. Printing is almost instantaneous on a PCL-5 or above printer. The computer's date and time at the time of printing appear in the title block. Date and time printing may be suppressed if desired (see page 11-8).

PRINT INTERMITTENT CX – (Print Intermittent Connections) prints a wiring graphic on your report printer showing any intermittent connections found. Intermittents appear as dashed lines, while other wiring appears as solid lines. This functions exactly like printing the intermittent connections manually from the intermittent connections wiring display screen. Because CONTINUOUS TEST generates the intermittent connections list, it must be executed first. If there are no intermittent connections, nothing is printed and execution continues without stopping. Printing requires about 15 seconds per page on a PCL-5 or above

printer. The computer's date and time at the time of printing appear in the title block. Date and time printing may be suppressed if desired (see page 11-8).

PRINT MATCH DATA LABEL – prints label text for the currently loaded Match Data on your label printer. Be sure that you set the label format as desired before printing (read about label preferences starting on page 11-10). If no label text exists, an error tone sounds and the Macro continues without stopping. *Important:* you will print *one row* of labels each time this instruction executes. Thus, if your labels are four-across, four labels will print *each time* the instruction executes.

PRINT TEST DATA LABEL – (same as PRINT MATCH DATA LABEL, except that Test Data label text is printed). To use this instruction in a Macro, you must manually enter Test Data label text before starting the Macro. If you execute Test Cable within the Macro, previously entered Test Notes or Label text will be cleared.

6.7 Other Information about Macros

Run-Time Errors: Minor errors that occur during Macro execution sound an error tone and may generate a message, but do not halt execution. Errors like this usually result from missing data (such as trying to compare Test and Match data when Match data is missing), and may correct themselves on successive loops.

Serious errors sound an error tone and halt execution, usually with a message. Such errors result from a defect in logic (such as branching to a nonexistent line) or a missing or duplicate file (such as loading a cable file that does not exist). If you test newly created Macros thoroughly before employing them in production, this should never happen. Don't forget about the STEP function (Macro menu) – it lets you check the result of each Macro instruction step-by-step to ensure that it operates correctly.

Infinite Loops: Many Macros are designed to operate indefinitely in an endless loop. In these cases, the operator must click "Stop" to halt execution. Macros that repeat indefinitely should include at least one WAIT FOR PB instruction so that you may cue the system when you are ready to start the next test cycle. If you print labels in batches and wish to loop without WAIT FOR PB, remember that in most cases the computer can loop faster than the printer can print, and the communications buffer in the printer can accumulate many unprinted labels that will continue to output long after you stop the Macro.

Deleting a Macro: There is no Delete Macro function in the Macro window. To remove a Macro from the list, simply drag it out of the "Macros" folder. If you have Macros you may wish use at a later time but do not want to appear in the list currently, create a folder called "Inactive Macros" and drag them there. To permanently delete a Macro, drag it to the trash.

Loading a Macro at Startup: The window setup that exists when you close the CableEye application will be reinstated when you start again. So, if you close the application with the Automatic Test window visible, it will reappear in exactly the same configuration with the same Macro selected.


To have CableEye start when you turn on your computer, first open "My Computer" on your desktop. Under "Scheduled Tasks", set up the CableEye program to open "At System Startup".



Using Shortcuts to Start Specific Macros: You may create individual shortcuts for CableEye, each with its own properties. One property you may customize defines which Macro loads when you start the software. You may thus create different shortcuts, each loading a different Macro at startup, to address a different job or test requirement. This eliminates the need for an unskilled operator to make a Macro choice when using CableEye and simplifies operation.

To set this up, create a shortcut for the CableEye executable (WinCi.exe) and right-click the shortcut. Choose "Properties" and click the "Shortcut" tab to reveal the window seen at the right. In the "Target" text box, after the executable's name, enter a space and then type the full name of the Macro you wish to load.

If desired, assign a hotkey in "Shortcut Key". In this case, pressing F5 starts CableEye and loads "Example2.mac". Once the software opens, pressing the TEST button on the fixture is all it takes to start execution.



6.8 Macro Examples

Nine examples are offered to illustrate some typical Macros and demonstrate how to apply the special control commands. You will find exactly these Macros installed with your system and may wish to try them!

Example 1: Before starting this Macro, you should manually load a Match Data file from the cable database corresponding to the cable you wish to test.

```
1  WAIT FOR PB
2  TEST CABLE
3  COMPARE TEST TO MATCH
4  IF MATCH THEN ...
5      BEEP
6  IF NOMATCH THEN ...
7      BEEP BEEP
8  REPEAT 1
```

Explanation of Example 1: (Line 1) Once started, we mount a cable and press the TEST pushbutton to continue. (2) Acquire Test Data and (3) compare it to the Match Data loaded earlier. (4) If they match, (5) sound a "beep" tone, whereas (6) if they do not match, (7) sound a "beep beep" tone. (8) Loop back to line 1 where we again wait for the TEST pushbutton. When a new cable has been mounted, press the pushbutton to repeat the test. When all cables have been tested, click "Stop" to end this Macro. If you wish the Difference List to appear after COMPARE TEST TO MATCH, either use the DISPLAY DIFFERENCE LIST command, which is not done here, or click the "triangle" button before starting the Macro. Remember that if you have a sound card installed on your computer, BEEP will create an *asterisk* sound, and BEEP BEEP will create a *critical stop sound*. The actual sound you hear depends on the .WAV assignment in your "Sounds" control panel.

Example 2: Before starting, you should mount a model cable on the test fixture and set up your printer.

```
1  LEARN CABLE
2  BEEP
3  WAIT FOR PB
```

```
4 TEST CABLE
5 COMPARE TEST TO MATCH
6 IF MATCH THEN ...
7     BEEP
8 IF NOMATCH THEN ...
9     BEEP BEEP
10    DISPLAY DIFFERENCE LIST
11 REPEAT 3
```

Explanation of Example 2: (Line 1) Acquire test data from the model cable and store it in the Match Data buffer. (2) Sound a tone to acknowledge completion. (3) While the system waits for the pushbutton to be pressed, remove the model cable and mount a test cable. Then, press the pushbutton to proceed. (4) Acquire Test Data from the cable under test. (5) Compare Test Data and Match Data. (6,7) If they match, sound a single tone. (8,9,10) If they do not match, sound a double tone and display the wiring differences. (11) Repeat the test from line 3.

Example 3:

```
1 WAIT FOR PB
2 CONTINUOUS TEST
3 IF NOMATCH THEN ...
4     BEEP BEEP
5     DISPLAY INTERMITTENT CX
6 REPEAT 1
```

Explanation of Example 3: After the Macro starts, mount a cable that is to be checked for intermittent connections. (Line 1) Press TEST to start the procedure. (2) CONTINUOUS TEST operates indefinitely until you press the pushbutton again. While the test is running, flex the cable at various points to check the integrity of internal wiring. When you flex the cable, be sure to hold the connector nearest the flex point firmly down so that it does not work loose from the test fixture. If changes in the continuity occur during flexing, short beeps will sound and the red "Fail" LED will turn on. (3) When you press the pushbutton to stop the continuous test, check for errors. If there were intermittent connections during the test, (4) sound a double tone, (5) then display intermittent connections. If no intermittent connections were found, the green "Pass" LED will turn on. (6) Repeat the procedure from line 1. At this time, you may remove the cable and mount another.

Example 4:

```
1  LOAD...DB25M-DB25F-S25D
2  WAIT FOR PB
3  TEST CABLE
4  COMPARE TEST TO MATCH
5  IF MATCH THEN ...
6      BEEP
7      REPEAT 2
8  SEARCH DISK FOR MATCH
9  IF MATCH THEN ...
10     PRINT MATCH DATA SPECS
11     REPEAT 1
12 BEEP BEEP
13 STOP
```

Explanation of Example 4: (Line 1) Load a specific cable file from the database. (2) Pause while you mount a cable. When ready, press the TEST pushbutton. (3) Measure the cable. (4,5) If Test Data and Match Data agree, (6) sound a single tone and (7) repeat from line 2. (8) If Test Data and Match Data disagree, search the database for a match. (9) If a match is found, then (10) print the Match Data wiring, netlist, notes, and label and (11) restart from line 1 to reload the original cable file. If the database does not contain a match, (12) sound a double tone and (13) end the Macro.

Comment: You may use a footswitch instead of the TEST pushbutton to trigger a Macro. We can provide a standard footswitch (catalog Item 714) that connects to the "Remote" socket on M2. You may also build your own footswitch according to wiring specifications for the remote control socket given earlier in this User's guide. A relay closure will also work if you are interfacing CableEye to automatic test equipment.

Example 5:

```
1 LEARN CABLE
2 WAIT FOR PB
3 TEST CABLE
4 COMPARE TEST TO MATCH
5 IF MATCH THEN...
6     BEEP
7 IF NOMATCH THEN...
8     BEEP BEEP
9     DISPLAY DIFFERENCE LIST
10 CONTINUOUS TEST
11 IF MATCH THEN...
12     BEEP
13 IF NOMATCH THEN...
14     BEEP BEEP
15     DISPLAY INTERMITTENT CX
16     PRINT INTERMITTENT CX
17 REPEAT 2
```

Explanation of Example 5: Like Example 2, we learn a model cable attached to the fixture when started, and then are ready to begin testing. In addition to checking for wiring equivalence, we also check here for intermittent connections starting on Line 10. Note that your Macros may easily perform multiple tests in one loop.

Example 6:

```
1 LOAD...DB25M-DB25F-S25D
2 PRINT MATCH DATA LABEL
3 IF COUNT = 8
4     STOP
5 REPEAT 2
```

Explanation of Example 6: This Macro loads a specific data file and prints 8 labels based on the content of the Match Data's label text. It runs automatically and stops when complete. Note that CableEye will print *eight rows* of labels, so if the labels are one-wide, you will get eight labels. However, if they are four wide, you will get 32 labels.

Example 7:

```
1  LOAD...CN50M-CN50F-S50D
2  SET INITIAL COUNT = 230
3  WAIT FOR PB
4  TEST CABLE
5  COMPARE TEST TO MATCH
6  IF MATCH THEN...
7      BEEP
8      PRINT MATCH DATA LABEL
9  IF NOMATCH THEN...
10     BEEP BEEP
11     SKIP CABLE COUNT
12 IF COUNT = 238
13     STOP
14 REPEAT 3
```

Explanation of Example 7: The label text for this cable file (CN50M-CN50F-S50D) includes the placeholder "<count>" embedded in the text; look this up in your standard cable database to find the following label text:

```
CN50M-CN50F, Shielded, 50-Conductor
Serial: BZ<count>
```

The run-time value of the count variable at the time the label is printed replaces "<count>" during execution. So, the first time we measure a good cable with this Macro, the actual printed label reads as follows:

```
CN50M-CN50F, Shielded, 50-Conductor
Serial: BZ230
```

The Macro stops after we complete nine cables (230 through 238 inclusive). Note that defective cables freeze the loop index (Line 11) so we do not skip any serial numbers.

Example 8:

```
1  ENTER CABLE NAME
2  ENTER INITIAL COUNT
3  WAIT FOR PB
4  TEST CABLE
5  COMPARE TEST TO MATCH
6  IF MATCH THEN...
7      PRINT MATCH DATA LABEL
8  IF NOMATCH THEN...
9      BEEP BEEP
10     SKIP CABLE COUNT
11 IF COUNT = 25
12     STOP
13 REPEAT 3
```

Explanation of Example 8: We assume here that the label text for this cable file (which you choose) includes the placeholder "<count>" embedded in the text. You will also set the initial count to an appropriate value. As in the previous example, the run-time value of the count variable at the time the label is printed replaces "<count>" during execution.

The Macro stops when the count value reaches 25. Note that if you initialize the count value to a number greater than 25, the Macro will never stop by itself. As usual, you may click the "Stop" button to halt it.